20

СЛАЙД 1
Для написання, налагодження та завантаження прошивок необхідно завантажити, встановити та налаштувати середовище розробки Arduino IDE.
Коли у нас є плата мікроконтролера та встановлене середовище розробки на комп’ютері, можна переходити до створення скетчу (прошивки).
Для програмування застосовується спрощена версія мови C++ із уже визначеними функціями.
Будь-яка прошивка повинна містити щонайменше дві функції: setup() і loop().
Функція setup() виконується на початку роботи програми лише один раз  одразу після ввімкнення або перезавантаження пристрою. Зазвичай у ній визначають режими роботи пінів, ініціалізують необхідні протоколи зв’язку, встановлюють з’єднання з додатковими модулями та налаштовують підключені бібліотеки (приклад 1).

#define LED_PIN 2

void setup() {
  Serial.begin(115200);
  pinMode(LED_PIN, OUTPUT);

  // Коротка пауза, щоб Serial Monitor встиг підключитися
  delay(2000);

  Serial.println("Система запустилась! Початкове налаштування виконано.");
}

У скетчах для Arduino / ESP32 функція loop()  це основна частина програми, яка повторюється безперервно від моменту запуску пристрою й до його вимкнення або перезавантаження.
Її можна уявити як "нескінченний цикл роботи мікроконтролера" (приклад 2).

void loop() {
  digitalWrite(LED_PIN, HIGH);
  Serial.println("Світлодіод увімкнено");
  delay(1000);

  digitalWrite(LED_PIN, LOW);
  Serial.println("Світлодіод вимкнено");
  delay(1000);
}

СЛАЙД 2
Serial.println()
Serial.println() – це функція Arduino/ESP32 для виводу даних через серійний порт (USB або UART) на комп’ютер.
· Дані потрапляють у Serial Monitor Arduino IDE.
· println означає “print line” – після виведення автоматично додається символ нового рядка (\r\n), щоб текст відображався на окремому рядку.

Синтаксис
Serial.println(data);
Serial.println(data, format);

	Параметр
	Опис

	data
	Дані для виводу. Може бути: число, текст (String або "текст"), символ, змінна.

	format
	(необов’язково) формат виводу чисел: BIN, OCT, DEC, HEX, або для плаваючих – кількість знаків після коми.



pinMode()

pinMode()  це вбудована функція, яка визначає режим роботи піна (GPIO).
Тобто ми вказуємо мікроконтролеру, як саме використовувати певний контакт:
· як вхід (INPUT),
· як вихід (OUTPUT),
· або як вхід із внутрішнім підтягувальним резистором (INPUT_PULLUP).
Синтаксис
pinMode(pin, mode);
· pin – номер піна (наприклад 2, 4, 13).
· mode – режим роботи (INPUT, OUTPUT, INPUT_PULLUP).
Режими роботи
1. INPUT (Вхід)
Пін працює для читання сигналу (наприклад, від кнопки чи датчика).
· Значення читають через digitalRead(pin).
· Якщо вхід "плаває" (нічого не підключено), може зчитуватися випадкове значення.
2. OUTPUT (Вихід)
Пін працює для керування (наприклад, світлодіодом, реле, мотором).
· Стан встановлюють через digitalWrite(pin, HIGH (1)/LOW (0)).
3. INPUT_PULLUP (Вхід із підтягуванням)
Вмикає внутрішній підтягувальний резистор (≈ 10 кОм до VCC).
· Корисно для кнопок: коли вони не натиснуті  пін читає HIGH, а при натисканні  LOW.

digitalRead()

digitalRead()  це функція, яка дозволяє зчитати логічний стан (0 або 1) з цифрового входу (GPIO).
· Якщо на піні є напруга 3.3В → читається HIGH (логічна 1).
· Якщо на піні 0В (GND) → читається LOW (логічний 0).

Синтаксис
int state = digitalRead(pin);
· pin – номер GPIO, налаштований як INPUT або INPUT_PULLUP у pinMode().
· Функція повертає:
· HIGH (1),
· LOW (0).

digitalWrite()
digitalWrite() – це функція, яка встановлює цифровий стан (HIGH або LOW) на виводі (піні), попередньо налаштованому як OUTPUT у pinMode().
Інакше кажучи, вона дозволяє вмикати або вимикати сигнал на піні.

Синтаксис
digitalWrite(pin, value);
· pin – номер GPIO (наприклад, 2, 4, 13).
· value – стан піна:
· HIGH – подати логічну 1 (3.3В на ESP32).
· LOW – подати логічний 0 (0В, заземлення).

Приклад використання функцій 
#define BUTTON_PIN 4    // GPIO4 - кнопка
#define LED_PIN 2       // GPIO2 - вбудований світлодіод

void setup() {
  // Налаштовуємо кнопку як вхід із підтягувальним резистором
  pinMode(BUTTON_PIN, INPUT_PULLUP);

  // Налаштовуємо світлодіод як вихід
  pinMode(LED_PIN, OUTPUT);
}

void loop() {
  // Зчитуємо стан кнопки
  int buttonState = digitalRead(BUTTON_PIN);

  // Якщо кнопка натиснута (LOW через INPUT_PULLUP)
  if (buttonState == LOW) {
    digitalWrite(LED_PIN, LOW);   // Увімкнути LED (на ESP32 логіка інверсна)
  } else {
    digitalWrite(LED_PIN, HIGH);  // Вимкнути LED
  }
}
Офіційна документація Arduino https://docs.arduino.cc/language-reference/
СЛАЙД 3

Тема: Датчики. Актуатори. Контролери
Вступ
Сучасні вбудовані системи, робототехніка, «розумні» пристрої та автоматизовані комплекси неможливі без взаємодії із зовнішнім середовищем. Для цього використовуються датчики, які сприймають інформацію, актуатори, які виконують дію, та контролери, що керують усією системою. Розуміння принципів їх роботи є основою для розробки систем керування, автоматизації та Інтернету речей (IoT).
1. Датчики
Датчик (сенсор)  пристрій, що перетворює фізичну величину (температуру, тиск, вологість, світло, звук, рух тощо) у електричний сигнал, який може бути оброблений контролером.
Основні види датчиків:
1. Датчики температури і вологості
· DHT11, DHT22  це цифрові датчики для вимірювання температури та відносної вологості повітря.
Вони мають вбудований мікроконтролер, який виконує оцифровку сигналів і передає готові дані у цифровому вигляді на ESP32.

[image: https://www.mini-tech.com.ua/image/cache/catalog/sensors/DHT11_3-550x550.png]
Рисунок 1  Датчики температури і вологості DHT11

Характеристики
	Характеристика
	DHT11
	DHT22 (AM2302)

	Діапазон температур
	0 … +50 °C
	–40 … +80 °C

	Точність температури
	±2 °C
	±0.5 °C

	Діапазон вологості
	20 … 80 %
	0 … 100 %

	Точність вологості
	±5 %
	±2–3 %

	Інтервал вимірювання
	1 раз/секунда
	0.5 секунди

	Напруга живлення
	3.3 … 5 В
	3.3 … 5 В

	Інтерфейс
	Однопровідний цифровий (Single-Wire)
	Однопровідний цифровий (Single-Wire)



Виводи:
1. VCC (3-5в живлення)
2. Data Out - вивід даних
3. NC - не використовується
4. Загальний
При підключенні до мікроконтролера, ви можете між виводами Vcc і Data розмістити підтягуючий pull-up резистор номіналом 10 кОм. 

· Простий цифровий датчик, передає дані по власному протоколу.
· DHT22 точніший і має більший діапазон, ніж DHT11.
· Використовується у клімат-контролі, метеостанціях.

СЛАЙД 4

Приклад  4:
#include "DHT.h"

// Налаштування
#define DHTPIN 4      // Пін ESP32, до якого підключено DATA DHT11 (GPIO4)
#define DHTTYPE DHT11 // Тип датчика: DHT11 або DHT22

DHT dht(DHTPIN, DHTTYPE);

void setup() {
  Serial.begin(115200);
  dht.begin();
  Serial.println("Запуск DHT11...");
}

void loop() {
  // Затримка між вимірюваннями (рекомендовано мінімум 2 секунди для DHT11)
  delay(2000);

  // Зчитування вологості
  float h = dht.readHumidity();
  // Зчитування температури у °C
  float t = dht.readTemperature();
  // Зчитування температури у °F
  float f = dht.readTemperature(true);

  // Перевірка чи вдалося отримати дані
  if (isnan(h) || isnan(t) || isnan(f)) {
    Serial.println("Помилка зчитування з датчика DHT11!");
    return;
  }

  // Вивід результатів
  Serial.print("Вологість: ");
  Serial.print(h);
  Serial.print(" %\t");

  Serial.print("Температура: ");
  Serial.print(t);
  Serial.print(" °C ");
  Serial.print(f);
  Serial.println(" °F");
}

СЛАЙД 5

· DS18B20


[image: https://arduino.ua/products_pictures/large_ds18b20-dallas.jpg]
Рисунок 2  Датчики температури DS18B20

· Діапазон вимірюваних температур від -55 до +125 °C. Зчитуваний з приладу цифровий код є прямим безпосереднім кодом виміряного значення температури і не потребує додаткових перетворень. Програмована користувачем роздільна здатність вбудованого АЦП може бути змінена в діапазоні від 9 до 12 розрядів вихідного коду. Абсолютна похибка перетворення менше 0.5 °C в діапазоні контрольованих температур -10 до +85 °C. Максимальний час повного 12-ти розрядного перетворення ~750 мс (при роздільній здатності 12 розрядів). Для підключення потрібно резистор 4.7 кОм.
· Внутрішня енергонезалежна пам'ять температурних установок забезпечує запис довільних значень верхньої та нижньої межі установок. Крім того, мікросхема містить вбудований логічний механізм пріоритетної сигналізації в лінію про факт виходу температури за один з обраних порогів. Вузол 1-Wire-інтерфейсу приладу організований таким чином, що існує теоретична можливість адресації необмеженої кількості подібних пристроїв на однопровідній лінії.
· Термометр має індивідуальний 64-розрядний реєстраційний номер (груповий код 028Н) і забезпечує можливість роботи без зовнішнього джерела живлення, тільки за рахунок паразитного живлення однопровідної лінії. Живлення приладу через окремий зовнішній вивід здійснюється напругою від 3.0В до 5.5В.

Застосовується у системах вимірювання температури рідин, ґрунту.

2. Датчики освітленості
· Фоторезистор (LDR)

Фоторезистор (LDR, Light Dependent Resistor)  це пасивний елемент, опір якого змінюється залежно від рівня освітленості:
· при яскравому світлі опір зменшується (кілька сотень Ом),
· у темряві  зростає (десятки-сотні кОм).
Його можна використати для вимірювання освітленості через ADC (аналогово-цифровий перетворювач) ESP32.

Схема підключення LDR до ESP32

1. Один вивід підключіть до +5В (живлення NodeMcu Esp32).
1. Інший вивід підключіть до аналогового входу NodeMcu Esp32, наприклад, GPIO32.
Резистор (наприклад, 10 кОм)
2. Один вивід підключіть до того ж аналогового входу GPIO32, до якого підключено фоторезистор.
2. Інший вивід підключіть до GND(земля) NodeMcu Esp32.
Як це працює 
· Фоторезистор змінює свій опір залежно від освітленості, а резистор утворює подільник напруги.
· NodeMcu Esp32 вимірює напругу на аналоговому вході GPIO32, яка залежить від співвідношення опорів фоторезистора та резистора.
· Чим менше світла, тим вищий опір фоторезистора, тим нижча напруга на GPIO32, і навпаки.

Рисунок 3  Схема підключення LDR до ESP32

Використання: автоматичне ввімкнення світла, системи «день/ніч».

СЛАЙД 6

Приклад (5) коду:
// Вибір ADC-піна (GPIO34 на ESP32, тільки вхід!)
const int LDR_PIN = 34;  

void setup() {
  Serial.begin(115200);
  delay(1000);
  Serial.println("Читання LDR (фоторезистор)...");
}

void loop() {
  // Зчитування аналогового значення (0...4095 для ESP32)
  int ldrValue = analogRead(LDR_PIN);

  // Перетворення в напругу (3.3V = 4095)
  float voltage = ldrValue * (3.3 / 4095.0);

  Serial.print("LDR (ADC): ");
  Serial.print(ldrValue);
  Serial.print(" | Напруга: ");
  Serial.print(voltage);
  Serial.println(" V");

  delay(1000);
}

СЛАЙД 7

· BH1750

BH1750FVI  це цифровий сенсор освітленості (ambient light sensor), який перетворює рівень освітленості на цифрове значення (в люксах), передаючи його через інтерфейс I²C bus (послідовний інтерфейс обміну даними). Він широко використовується для автоматичного регулювання яскравості дисплеїв, клавіатурної підсвітки, в агродослідженнях та інших проектах.
Ключові характеристики BH1750
· Цифровий вихід у люксах (lx)  забезпечує 16-бітні вимірювання, від 1 до 65,535 lx, з можливістю розширення до ≈100,000 lx при калібруванні.
· Інтерфейс I²C  працює у двох режимах (fast-/standard), має режим низького енергоспоживання. 
· Спектральна чутливість близька до людського ока.
· Відносно мала вплив інфрачервоного випромінювання, стійкість до типів освітлення (лампи розжарення, денне світло тощо). 
· Функція відсічення "мерехтіння" при 50 / 60 Гц, що зменшує шум при штучному світлі. 
· Низьке споживання струму з можливістю переходу в режим Power-down.
· Не вимагає додаткових зовнішніх компонентів, готовий до підключення. 
Як працює сенсор
BH1750 складається з фотодіода, який реагує на світло, після чого сигнал підсилюється, перетворюється в цифровий формат (A/D вбудовано) і передається через I²C. 

Режими вимірювання
· Continuous Mode (Безперервний режим)  сенсор безперервно вимірює освітленість.
· One-Time Mode (Одноразовий режим)  один вимір і перехід до енергозберігаючого стану.
· Три рівні точності:
· Низька (Low Res)  чутливість ~4 lx, швидке опитування (~16 мс)
· Висока (High Res)  ~1 lx точність (~120 мс)
· Висока 2 (High Res2)  ~0.5 lx точність (~120 мс). 
Типові сфери застосування:
· Автоматичне регулювання яскравості дисплеїв, підсвітки
· Визначення денного чи нічного режиму
· Розумне освітлення, домашня автоматизація
· Лічильники освітленості
· Проекти з Arduino, ESP та інші мікроконтролерні системи
Підключення BH1750 до NodeMCU ESP32
	BH1750
	ESP32 (NodeMCU)
	Опис

	VCC
	3V3
	Живлення 3.3 В

	GND
	GND
	Земля

	SDA (Serial Data) Це лінія даних інтерфейсу I²C.  По ній передаються біти інформації між мікроконтролером (ESP32) та периферійним пристроєм (наприклад, BH1750).
	GPIO21 (SDA)
	Лінія даних I²C

	SCL Це лінія тактових імпульсів (годинника).
Вона керує тим, коли зчитується/записується кожен біт на SDA.
Майстер (ESP32) генерує ці імпульси
	GPIO22 (SCL)
	Лінія годинника I²C

	ADDR (Address Select)
Це спеціальний вивід у деяких датчиків (наприклад, BH1750).
Він визначає I²C-адресу пристрою, щоб можна було підключати кілька однакових датчиків на одну шину I²C.

	GND (або VCC)
	Вибір I²C-адреси



На ESP32 SDA і SCL можна переназначати, але за замовчуванням SDA = GPIO21, SCL = GPIO22.

СЛАЙД 8

Рисунок 4  Схема підключення BH1750FVI до ESP32

Приклад коду 6:
#include <Wire.h> Wire, яка використовується для роботи з шиною I2C (двопровідний інтерфейс обміну даними між мікроконтролером і пристроями)
#include <BH1750.h>

// Створюємо об'єкт датчика
BH1750 lightMeter;

void setup() {
  Serial.begin(115200);

  // Ініціалізація I2C (ESP32: SDA = GPIO21, SCL = GPIO22)
  Wire.begin(21, 22);

  // Ініціалізація BH1750 у режимі безперервних вимірювань з високою роздільністю
  if (lightMeter.begin(BH1750::CONTINUOUS_HIGH_RES_MODE)) {
    Serial.println("BH1750 успішно ініціалізовано!");
  } else {
    Serial.println("Помилка ініціалізації BH1750. Перевірте з'єднання!");
  }
}

СЛАЙД 10

void loop() {
  // Зчитуємо значення освітленості в люксах
  float lux = lightMeter.readLightLevel();

  Serial.print("Освітленість: ");
  Serial.print(lux);
  Serial.println(" lx");

  delay(1000); // Оновлення раз на 1 сек.
}

СЛАЙД 11

3. Датчики руху та присутності
· PIR-датчик (HC-SR501)
PIR (Passive InfraRed) датчик  це сенсор руху, який виявляє зміну інфрачервоного випромінювання від об’єктів (наприклад, людини чи тварини).
Він не випромінює нічого сам, а лише реєструє різницю теплового фону.
HC-SR501  один з найбільш поширених та дешевих PIR-датчиків.
Основні характеристики HC-SR501
· Напруга живлення: 4.5  20 В (зручно підключати до 5 В).
· Споживання струму: ≈ 50 мкА (дуже економний).
· Робоча температура: –15…+70 °C.
· Дальність виявлення: 3  7 м (регулюється).
· Кут виявлення: приблизно 120° (залежить від лінзи).
· Час спрацювання: регулюється від ≈ 0.3 до 300 секунд.
· Вихід: цифровий сигнал (HIGH = рух виявлено, LOW = руху немає).
Конструкція та регулювання
На платі HC-SR501 є:
1. ІЧ-сенсор з білою лінзою Френеля (розсіює теплове випромінювання, збільшує зону огляду).
2. Два потенціометри:
· Чутливість (відстань)
· Час утримання сигналу після виявлення
3. Перемикач режиму роботи (H/L):
· H (retriggerable): датчик буде тримати вихід HIGH, поки рух триває.
· L (non-retriggerable): датчик подасть імпульс лише один раз, навіть якщо рух продовжується.
Схема підключення (ESP32)
	Вивід
	Опис

	VCC
	Живлення 5 В (може працювати і від 3.3 В на ESP32)

	GND
	Земля

	OUT
	Цифровий вихід (підключається до GPIO)

Рисунок 5  Схема підключення HC-SR501 до ESP32

Як працює
1. У спокої датчик запам’ятовує інфрачервоний фон.
2. Якщо в полі зору з’являється теплий об’єкт (людина, тварина), ІЧ-випромінювання змінюється.
3. Датчик подає сигнал HIGH на вихідний пін (OUT).
4. Після закінчення часу затримки OUT знову стає LOW.
Використання: охоронні системи, автоматичне освітлення.
Приклад коду 7:
int ledPin = 13;
int switchPin = 2;
int value = 0;
void setup() {
	pinMode(ledPin, OUTPUT);
	pinMode(switchPin, INPUT);
}
void loop() {
	value = digitalRead(switchPin); 
	if (HIGH == value) {
		digitalWrite(ledPin, HIGH);
	} else {
		digitalWrite(ledPin, LOW); 
	}
}

4. Датчики відстані
· HC-SR04 (ультразвуковий датчик)
HC-SR04  це популярний ультразвуковий датчик відстані, який використовує принцип відбиття ультразвукових хвиль.
Він вимірює час, за який сигнал відбивається від об’єкта, і розраховує відстань.
Основні характеристики
· Живлення: 5 В
· Споживання струму: ~15 мА
· Дальність вимірювання: від 2 см до 400 см
· Роздільна здатність: ≈ 3 мм
· Кут огляду: ~15°
· Інтерфейс: цифровий (вхід/вихід)
Як працює
1. На Trig подається короткий імпульс (10 мкс).
2. Датчик генерує серію ультразвукових імпульсів (40 кГц).
3. Якщо перед датчиком є об’єкт, хвиля відбивається назад.
4. На виході Echo з’являється сигнал HIGH, тривалість якого пропорційна відстані.
5. Мікроконтролер вимірює час імпульсу й обчислює дистанцію:
Формула:
                                              (1)
де:
· Time – час сигналу (мікросекунди, мкс), отриманий з виводу Echo,
· 2 – ділиться на два, бо звук проходить туди і назад (до об’єкта і назад),
· SpeedOfSound – швидкість звуку у повітрі (~343 м/с = 0.0343 см/мкс).

Приклад застосувань
· Роботи-навігатори (уникнення перешкод).
· Автоматичні парктроніки.
· Системи вимірювання рівня рідини.
· Автоматичні двері та охоронні системи.
[image: ]
Рисунок 6  Схема підключення HC-SR04 до ESP32

Приклад коду 8:
#define TRIG_PIN 5   // GPIO5
#define ECHO_PIN 18  // GPIO18

long duration;
float distance;

void setup() {
  Serial.begin(115200);
  pinMode(TRIG_PIN, OUTPUT);
  pinMode(ECHO_PIN, INPUT);
}

void loop() {
  // Генеруємо імпульс на Trig
  digitalWrite(TRIG_PIN, LOW);
  delayMicroseconds(2);
  digitalWrite(TRIG_PIN, HIGH);
  delayMicroseconds(10);
  digitalWrite(TRIG_PIN, LOW);

  // Вимірюємо тривалість сигналу
  duration = pulseIn(ECHO_PIN, HIGH);

  // Обчислюємо відстань (швидкість звуку ~343 м/с)
  distance = (duration * 0.0343) / 2;

  Serial.print("Відстань: ");
  Serial.print(distance);
  Serial.println(" см");

  delay(500);
}

· VL53L0X (лазерний ToF-сенсор)
VL53L0X V2  це вдосконалена версія лазерного датчика відстані сімейства Time-of-Flight (ToF) від STMicroelectronics. Він вимірює відстань шляхом випромінювання невидимого інфрачервоного лазерного імпульсу (VCSEL-діод) і підрахунку часу його повернення після відбиття від об’єкта.
Це набагато точніше, ніж у звичайних ультразвукових або ІЧ-сенсорів, адже вимірювання ґрунтується на часі польоту світла, а не на інтенсивності відбитого сигналу.

[image: https://arduino.ua/products_pictures/large_dat309_1.jpg]
Рисунок 7  Лазерний датчик 

Основні характеристики
· Принцип роботи: Time-of-Flight (час польоту світла)
· Діапазон вимірювання:
· до ~2 м у приміщенні (при нормальних умовах)
· до ~1 м на вулиці (залежно від освітлення та поверхні об’єкта)
· Точність: ±3 % (типове значення)
· Кут огляду (FoV): ~25°
· Швидкість вимірювання: до 50 Гц (20 мс на вимірювання)
· Довжина хвилі лазера: ~940 нм (невидиме інфрачервоне світло)
· Інтерфейс: I²C (адреса за замовчуванням 0x29, змінна через XSHUT)
· Живлення: 2.6 – 3.5 В (часто модулі мають вбудований стабілізатор для 5 В)
· Логічні рівні: 3.3 В (для SDA/SCL), тому при 5 В потрібні сумісні модулі з перетворювачем

Піни модуля VL53L0X
(типовий модуль на платі розширення)
· VIN – живлення (3.3–5 В, залежить від плати)
· GND – земля
· SDA (Serial Data  це лінія передачі даних у послідовному інтерфейсі I²C (Inter-Integrated Circuit) – I²C дані
· SCL (Serial Clock) – I²C тактова
· XSHUT – вимкнення/скидання (LOW = вимкнено, HIGH = працює). 
XSHUT – це спеціальний вивід (пін) у сенсора VL53L0X, який використовується для керування його живленням і адресою.
Його функції:
1. Апаратне вимкнення / вмикання сенсора
· Якщо подати LOW (0) → сенсор повністю вимикається (споживає ~0).
· Якщо подати HIGH (1) → сенсор запускається і готовий працювати.
2. Призначення I²C-адреси (коли кілька датчиків на одній шині)
· За замовчуванням усі VL53L0X мають однакову адресу 0x29.
· Якщо підключити кілька таких модулів, вони будуть «конфліктувати» на шині I²C.
· Рішення:
· Тримаємо всі сенсори вимкненими через XSHUT.
· Вмикаємо по одному, і кожному змінюємо адресу програмно.
· Після цього всі працюють одночасно з різними адресами.
Як підключати
· Якщо у тебе лише один датчик – XSHUT можна не використовувати (залишити не підключеним, на модулі зазвичай він вже підтягнутий до HIGH).
· Якщо декілька датчиків – обов’язково підключати XSHUT кожного до окремого GPIO ESP32.

· GPIO1 / INT – переривання (опційно)

Особливості
· Покращений алгоритм вимірювання в умовах слабкого відбиття (чорні, матові об’єкти).
· Трохи менший енергоспоживання у режимі очікування.
· Сумісність з тими ж бібліотеками, що і VL53L0X першої версії.
· Дальність у більшості модулів залишилася до 2 м, але стабільність і швидкість вимірювань краща.

Застосування
· Роботи та дрони (уникнення перешкод)
· Смарт-пилососи та побутова техніка
· Лазерні рулетки, вимірювачі відстані
· Лічильники присутності (люди, об’єкти)
· Сенсори жестів і наближення


5. Датчики тиску та висоти
· BMP180 / BMP280 / BME280
Датчики сімейства BMP/BME від Bosch Sensortec призначені для вимірювання атмосферного тиску та температури, а також, у випадку BME280, ще й відносної вологості.
На основі виміряного тиску можна обчислювати висоту над рівнем моря, що робить ці сенсори корисними у портативних пристроях, дронах, системах навігації та метеостанціях.
BMP180
· Старіша модель, але все ще часто використовується.
· Параметри:
· Діапазон тиску: 300–1100 гПа.
· Точність: ±1 гПа (~±8 м по висоті).
· Температура: від –40 °C до +85 °C.
· Інтерфейс: I²C  це послідовний двопровідний інтерфейс для зв’язку між мікроконтролером (ESP32) та периферійними пристроями.
· Живлення: 1.8–3.6 В.
· Особливості: проста інтеграція, мала споживана потужність, але немає вимірювання вологості.

СЛАЙД 17

BMP280
· Оновлена версія BMP180, більш точна й швидка.
· Параметри:
· Діапазон тиску: 300–1100 гПа.
· Точність: ±1 гПа (~±1 м по висоті).
· Температура: від –40 °C до +85 °C.
· Інтерфейси: I²C або SPI (це високошвидкісний послідовний інтерфейс обміну даними між мікроконтролером (ESP32) та периферією).
· Живлення: 1.8–3.6 В (модулі часто мають стабілізатор на 3.3/5 В).
· Особливості: нижче енергоспоживання, висока швидкість вибірки, підходить для дронів та мобільних застосунків.
[image: https://arduino.ua/products_pictures/large_ADC208-5.jpg]
Рисунок 8  Датчики тиску та висоти

BME280
· Розширена версія BMP280, з додатковим датчиком вологості.
· Параметри:
· Діапазон тиску: 300–1100 гПа.
· Точність: ±1 гПа (~±1 м по висоті).
· Вологість: 0–100 % RH (точність ±3 %).
· Температура: від –40 °C до +85 °C.
· Інтерфейси: I²C або SPI.
· Живлення: 1.8–3.6 В (часто плати мають 3.3/5 В).
· Особливості: універсальний сенсор для вимірювання кліматичних параметрів (тиск, температура, вологість), широко використовується у метеостанціях та «розумних» IoT-системах.

I²C підключення (рекомендовано)
Схема
· BMP280 VCC → 3.3V (якщо модуль має стабілізатор — можна і 5V, але краще 3.3V).
· BMP280 GND → GND.
· BMP280 SDA → GPIO21 (SDA).
· BMP280 SCL → GPIO22 (SCL).
· Якщо є пін CSB і SDO – їх можна залишити не підключеними (або CSB = HIGH для I²C).

Код (Arduino IDE, бібліотека Adafruit BMP280)    (9)
#include <Wire.h>
#include <Adafruit_Sensor.h>
#include <Adafruit_BMP280.h>

Adafruit_BMP280 bmp; // створюємо об’єкт датчика

void setup() {
  Serial.begin(115200);
  if (!bmp.begin(0x76)) {  // Адреса може бути 0x76 або 0x77
    Serial.println("Не знайдено BMP280!");
    while (1);
  }
}

void loop() {
  Serial.print("Температура = ");
  Serial.print(bmp.readTemperature());
  Serial.println(" *C");

  Serial.print("Тиск = ");
  Serial.print(bmp.readPressure() / 100.0F);
  Serial.println(" hPa");

  Serial.print("Висота = ");
  Serial.print(bmp.readAltitude(1013.25)); // стандартний тиск на рівні моря
  Serial.println(" м");

  Serial.println();
  delay(2000);
}

6. Газові та екологічні датчики
· MQ-серія (MQ-2, MQ-7, MQ-135 тощо)
Датчики MQ – це напівпровідникові газові сенсори на основі SnO₂ (діоксиду олова). Їх принцип роботи базується на зміні електричного опору чутливого шару при контакті з різними газами.
· Працюють від 5 В.
· Мають аналоговий вихід (AO) та у деяких моделях цифровий вихід (DO).
· Використовуються у системах моніторингу якості повітря, сигналізаціях, IoT-проєктах.
· Потребують підігріву сенсорного елемента для стабільної роботи (від кількох хвилин до 24 годин залежно від моделі).

MQ-2 – датчик диму та горючих газів
Призначення:
· Виявлення диму, LPG (пропан-бутан), метану, водню, спирту.
Характеристики:
· Напруга живлення: 5 В
· Час прогріву: ~20 секунд (повна стабільність – кілька хвилин)
· Чутливість: 300 – 10 000 ppm
· Використання:
· сигналізації про пожежу,
· витік газу в побутових приміщеннях,
· робототехніка (розпізнавання диму).
MQ-7 – датчик чадного газу (CO)
Призначення:
· Виявлення чадного газу (CO), який є небезпечним для людини.
Особливість:
· Працює у двох режимах нагріву (цикл високої та низької температури):
· при 5 В – відбувається очищення сенсора,
· при 1,5 В – проводиться вимірювання.
Характеристики:
· Діапазон: 10 – 10 000 ppm CO
· Напруга живлення: 5 В
· Використання:
· системи безпеки у будинках,
· сигналізація у гаражах,
· моніторинг вентиляції.
MQ-135 – датчик якості повітря
Призначення:
· Виявлення шкідливих газів у навколишньому середовищі:
· аміак (NH₃),
· діоксид сірки (SO₂),
· бензол,
· дим,
· CO₂ (опосередковано, не дуже точний).
Характеристики:
· Напруга живлення: 5 В
· Діапазон: 10 – 1000 ppm
· Використання:
· системи моніторингу повітря у містах,
· IoT-проєкти для "розумного будинку",
· екологічні станції.
Схема підключення MQ до ESP32 (38 pin)
1. VCC → 5V (ESP32 має вихід 5В з USB, можна підключити сюди).
2. GND → GND.
3. AO → будь-який аналоговий пін ESP32 (наприклад GPIO34, GPIO35, GPIO36, GPIO39 – тільки для читання, без digitalWrite).
4. (необов’язково) DO → будь-який цифровий пін ESP32 (наприклад GPIO32) – якщо потрібна проста логіка "є газ/нема газ".

6. Датчики руху та орієнтації
· GY-521 (MPU-6050)

[image: ]
Рисунок 9   Датчики руху та орієнтації GY-521

GY-521  це модуль на базі MPU6050, тобто він містить:
1. 3-осьовий акселерометр – вимірює прискорення по X, Y, Z.
2. 3-осьовий гіроскоп – вимірює кутову швидкість об’єкта по X, Y, Z.
Модуль GY-521 вже має регулятор напруги та рівень логіки, тому його легко підключати до ESP32 або Arduino.

Основні характеристики
· Напруга живлення: 3.3V – 5V (вбудований стабілізатор напруги)
· Інтерфейс: I²C
· Піни: VCC, GND, SDA, SCL, AD0, INT
· Акселерометр: ±2, ±4, ±8, ±16 g (g – це одиниця прискорення, що дорівнює прискоренню вільного падіння:
1 g ≈ 9,81 м/с².)
· Гіроскоп: ±250, ±500, ±1000, ±2000 °/с (°/с (градусів за секунду) – це швидкість обертання навколо осі.)
· Діапазон вимірювань: 16-бітний для кожної осі

Підключення GY-521 до ESP32 (38 pin)
	GY-521 (MPU6050)
	ESP32

	VCC
	3.3V

	GND
	GND

	SDA
	GPIO21 (SDA)

	SCL
	GPIO22 (SCL)

	AD0
	GND (залишити для адреси 0x68)

	INT
	не обов’язково (для переривань)


На ESP32 пін SDA/SCL можна змінювати, якщо потрібно, головне правильно вказати у Wire.begin(SDA, SCL).
Для чого використовується GY-521
· Визначення нахилу та орієнтації об’єкта
· Робототехніка та дрони
· Ігрові контролери та стабілізатори камер
· Будь-які проекти, де потрібен контроль руху або положення


8. Детектор напруги на оптопарі

Детектор напруги на оптопарі дозволяє безпечно контролювати мережеву або іншу високовольтну напругу через мікроконтролер.

Принцип роботи
1. Вхідна напруга (наприклад 220 В AC) через розділовий резистор або дільник живить світлодіод оптопари.
2. Світлодіод світиться, коли напруга є.
3. Фототранзистор на виході змінює стан, створюючи цифровий сигнал (HIGH/LOW).
4. Мікроконтролер читає цей сигнал через пін, налаштований як INPUT.


[image: Модуль/детектор/датчик змінної напруги 220В з оптопарою гальванічною розв'язкою - фото 1 - id-p1876082800] 
Рисунок 10   Детектор напруги на оптопарі

Підключення до ESP32
	Оптопара / Детектор
	ESP32

	Вихід (з фототранзистора)
	GPIO4 (або будь-який вхід)

	GND
	GND

	VCC (логічна сторона)
	3.3V або 5V, залежно від схеми



Сторона високої напруги підключається через резистор до світлодіода всередині оптопари.
Де використовується
· Контроль присутності 220 В у мережі
· Системи автоматизації та розумного будинку
· Безпечний моніторинг напруги без прямого контакту з високою напругою
· Реле з гальванічною ізоляцією

2. Актуатори
Актуатор (виконавчий механізм) – це пристрій, який отримує електричний сигнал від контролера (наприклад, ESP32) і перетворює його у фізичну дію:
· рух (обертальний чи поступальний),
· світло,
· звук,
· тепло,
· перемикання механізмів.
Іншими словами: датчики вимірюють, а актуатори виконують.
Основні типи актуаторів
1. Електромеханічні
Вони перетворюють електричну енергію в механічний рух.
· DC-мотори (двигуни постійного струму): Використовуються для простого обертання коліс, вентиляторів. Регулюються напругою або ШІМ (PWM).
· Сервоприводи: Дають можливість обертати вал на певний кут (зазвичай 0–180°). Використовуються в роботах, маніпуляторах.
· Крокові двигуни: Рухаються по кроках (наприклад, 1,8° за крок), що дозволяє точно контролювати позицію. Застосування: 3D-принтери, ЧПУ-станки.

2. Електромагнітні
Використовують магнітне поле для створення руху.
· Реле: Це електронні вимикачі, які можуть вмикати/вимикати інші пристрої (освітлення, двигуни, нагрівачі).
· Соленоїди: Це котушка з сердечником, яка при подачі струму рухається вперед-назад. Використовується у замках, клапанах, автоматичних механізмах.
3. Світлові та звукові
Перетворюють сигнал у візуальне чи звукове повідомлення.
· Світлодіоди (LED): Найпростіший актуатор – індикація стану.
· Дисплеї (LCD, TFT): Виводять текст, графіку, дані.
· Динаміки, бузери: Використовуються для подачі сигналів (наприклад, звук помилки чи будильник).
4. Інші
· Нагрівачі (резистивні елементи): Перетворюють електроенергію на тепло (підігрів датчиків, систем опалення).
· Клапани (електроклапани): Керують подачею рідини або газу. Використовуються у поливних системах, компресорах.
Приклади використання актуаторів у проектах:
· Робототехніка: мотори та серви рухають колесами та маніпуляторами.
· Розумний дім: реле вмикає лампи, клапани керують поливом.
· Смартфони: вібромотор для повідомлень.
· 3D-принтер: крокові двигуни задають точність руху головки.
· Сигналізація: бузер подає гучний звук при спрацюванні датчика.

3. Контролери
Контролер  це мікропроцесорний пристрій, який:
1. Зчитує дані від датчиків (температура, тиск, рух тощо).
2. Обробляє інформацію за закладеною програмою (алгоритмом).
3. Приймає рішення і керує актуаторами (мотор, світлодіод, клапан, реле тощо).
Тобто контролер  це "мозок" системи автоматизації чи IoT.

Основні характеристики контролерів
1.  Обчислювальна потужність
· Визначається частотою процесора (MHz, GHz) та кількістю ядер.
· Від цього залежить, наскільки швидко контролер зможе обробляти дані й виконувати алгоритми.
· Наприклад:
· ESP32  до 240 МГц, 2 ядра.

2. ОЗП (RAM)
· Використовується для збереження тимчасових даних: змінні, буфери, проміжні результати обчислень.
· Чим більше RAM, тим складніші алгоритми можна реалізувати.
· Наприклад:
· ESP32  520 КБ RAM.

3. Flash-пам’ять
· Тут зберігається програма (прошивка), яку ми завантажуємо в контролер.
· Також може використовуватися для постійних даних (наприклад, Wi-Fi налаштувань).
· Приклади:
· ESP32  від 4 МБ Flash.

4. Інтерфейси введення-виведення
Це "канали зв’язку", через які контролер спілкується із зовнішнім світом.
· GPIO (General Purpose Input/Output): універсальні цифрові входи/виходи (для кнопок, світлодіодів, датчиків).
· PWM (Pulse Width Modulation): керування яскравістю світлодіодів або швидкістю моторів.
· UART (послідовний порт): обмін даними між двома пристроями (наприклад, ESP32 ↔ комп’ютер).
· I²C (Inter-Integrated Circuit): шина для підключення кількох датчиків (наприклад, дисплей + акселерометр).
· SPI (Serial Peripheral Interface): швидкісний інтерфейс для пам’яті, дисплеїв, модулів.
· CAN (Controller Area Network): використовується в автомобілях для зв’язку між блоками управління.

Приклади популярних контролерів
· Arduino Uno / Mega: прості для навчання, робототехніки, невеликих проектів.
· ESP8266 / ESP32: мають Wi-Fi та Bluetooth, підходять для IoT (розумний дім, датчики).
· STM32: потужніші 32-бітні контролери для промислових і робототехнічних рішень.
· Raspberry Pi (умовно контролер): мікрокомп’ютер, який може виконувати роль контролера з ОС Linux.

Як працює контролер у системі "Датчик → Контролер → Актуатор":
1. Датчик передає дані (наприклад, "температура = 30 °C").
2. Контролер обробляє (наприклад, "якщо температура > 28 °C, увімкнути вентилятор").
3. Контролер подає сигнал на актуатор (реле вмикає вентилятор).

5. Застосування у практиці
· Розумний дім: керування освітленням, опаленням, сигналізацією.
· Промисловість: автоматизовані конвеєри, роботи-маніпулятори.
· Медицина: датчики пульсу, апарати штучного дихання.
· Автомобілі: системи ABS, парктроніки, круїз-контроль.
· IoT: «розумні» лічильники, екологічні сенсори.

Висновок
Датчики, актуатори та контролери є базовими складовими будь-якої сучасної автоматизованої системи. Взаємодія між ними дозволяє приймати рішення та впливати на низ. Розуміння принципів їх роботи відкриває шлях до створення власних проектів у галузі робототехніки, IoT та кіберфізичних систем.

